
Abstract

Roel de Vrijer’s assignment maps a simply typed λ-term to a measure whose numerical
component decreases with each β-contraction. We present that assignment as a compact
Haskell program in which object-language binders are implemented by Haskell binders, and
the functional component of the measure is stored as an ordinary Haskell function, so that
no separate environment-passing evaluator is needed in the main program. A proof that this
is a decreasing bound on all subsequent reduction sequences is included. An accompanying
Rocq development interprets intrinsically typed de Bruijn terms in the same measurement
algebra and machine-checks substitution, contextual decrease, and strong normalization.

1 Introduction
A challenge, attributed to Kurt Gödel and William Howard, is to find a transparent ordinal-
valued measure for simply typed terms that decreases every time a β-redex is contracted. The
open-problem list for rewriting includes this question [11, 9, Problem 19], and it likewise appears
as an “unsolved” problem in the typed λ-calculus (TLCA) list [13, Problem 26].

Strong normalization itself has many standard proofs, most notably via reducibility or related
syntactic arguments [17, 22, 12, 15, 14, 19]. Roel de Vrijer’s solution [8], further explained in an
unpublished note by Henk Barendregt [4], is significant because the size is computed together
with a functional continuation of the same estimate: The measure of a function term carries
along enough information to measure all future applications of that function term.

A Scheme implementation of de Vrijer’s measure was given at the termination workshop in
2021 [10]. Here it is recast in Haskell, a call-by-need language. Since the dot component can be
stored directly as a Haskell function, the implementation eliminates the quoting and unquoting
used in the Scheme version. Our implementation below goes further and lets Haskell binders
carry the object-language environment through ordinary lexical scope.

De Vrijer’s finite-developments theorem is an earlier example of such a quantitative analysis
of reduction: a natural-number annotation gives the length of a longest complete development [7].
Sørensen later treated the dual problem for shortest developments [20]. These results belong to
the same broad tradition of measuring reductions directly. Other decreasing measures for simply
typed β-reduction include Barenbaum and Sottile’s W - and Tm-measures [3] and Stirton’s proof-
theoretic natural-number assignment for λβ-reduction [21]. Schwichtenberg and Beckmann study
bounds on the length of typed reduction sequences [18, 5]. There is also an unpublished Agda
formalization of weak call-by-value reduction by Barenbaum [1]. Related uses of developments

E. Ábrahám, N. Bjørner and M. Rawson (eds.), LPAR-26 (EPiC Series in Computing, vol. 112), pp. 418–429



de Vrijer’s Strong Normalization in Haskell Nachum Dershowitz

and decreasing measures appear in Boudol’s intersection-type proof [6] and in later work by
Barenbaum, Ronchi Della Rocca, and Sottile on idempotent intersection types [2].

The code archive1 contains both this simple higher-order implementation and a named,
environment-passing first-order implementation; only the simple implementation is described
and proved correct here. It also contains a Rocq [23] (formerly Coq) development based
on intrinsically typed de Bruijn syntax. Its evaluator folds that syntax into the same type-
indexed algebra of measurements as the Haskell combinators. The development formalizes typed
renaming and substitution, full contextual β-reduction, strict decrease of the measure, and
strong normalization.2

2 The Measure
The construction has three layers. First, a measurement domain is assigned to every simple
type. Second, two auxiliary operations propagate a numerical charge through those domains
and provide canonical measurements at arbitrary types. Third, the three syntactic forms are
interpreted in the resulting algebra.

Measurement domains. Terms and types are as usual. For every type τ we have a domain
Dτ of semantic measurements, with Do = N and Dσ→τ = (Dσ → Dτ ) × N. A base-type
measurement is a number; a function-type measurement a = ⟨f, n⟩ has dot part a• = f and star
part a∗ = n. At base type, n∗ = n. The star part is the current reduction budget; the dot part
measures future applications. For example, ⟨f, 3⟩ ∈ Do→o assigns budget 3 to the unapplied
function and produces f(n) when applied to an argument measured by n.

Lifting charges and canonical measurements. The semantic clauses use two operations.
The first, addk, lifts the addition of k through a measurement of any type:

addk(n) = n+ k, addk(⟨f, n⟩) = ⟨λa. addk(f(a)), n+ k⟩.

The charge is added both to the current star component and recursively to every future result of
the dot component. This is essential at function type: a charge attached before an application
must still be present after that application. At type o → o, both the unapplied budget and
every base-type result increase by k.

The second operation gives the canonical measurement of star value n at a specified type:

co(n) = n, cσ→τ (n) = ⟨λa. cτ (n+ a∗), n⟩.

At type o → o this specializes to co→o(n) = ⟨λa. n+ a, n⟩. In particular, co→o(0) = ⟨λa.a, 0⟩.
Canonical measurements complete finitely written valuations and supply the probe for an
abstraction’s star component. Valuations are written as finite maps: an unlisted x has value
ρ(x) = ctype(x)(0), making the valuation total. For a closed term evaluated from the empty
map, every variable occurrence is supplied by its binder’s update. The probe cσ(0) measures an
abstraction’s body before an argument is supplied.

1At https://github.com/NachumDershowitz/SN_Haskell.
2ChatGPT 5.5 Pro assisted in developing the Haskell implementation, Rocq formalization, and proofs.

Interpretation of terms. Assuming that a valuation ρ maps each variable of type τ to an
element of Dτ , define

[[x]]ρ = ρ(x),

[[MN ]]ρ = [[M ]]•ρ([[N ]]ρ),

[[λxσ.M ]]ρ = ⟨λa. adda∗+1

(
[[M ]]ρ[x7→a]

)
, [[M ]]∗ρ[x7→cσ(0)]

⟩.

Here ρ[x 7→ a] denotes valuation update. Capture-avoiding substitution of N for x in M is
written M [x 7→ N ]. For a closed term M , the computed measure is ∥M∥ = [[M ]]∗∅.

The abstraction’s dot part evaluates the body with x 7→ a and adds a∗ + 1: 1 pays for the
outer β-contraction, while a∗ accounts for reductions in the argument even when the body ignores
it. Its star part records the body’s budget with x assigned its canonical zero measurement.

The central decrease is already visible at a root redex. If a = [[N ]]ρ, then

[[(λxσ.M)N ]]ρ = adda∗+1

(
[[M ]]ρ[x7→a]

)
. (1)

The substitution lemma proved in Section 4 identifies the term inside add with [[M [x 7→ N ]]]ρ.
Thus the redex has the measurement of its contractum plus a positive charge, lifted through the
result type.

As a preliminary example, let I = λxo.x. At a base measurement n, its dot part is
addn+1(n) = 2n+ 1 and its star part is 0, so

[[I]]∅ = ⟨λn. 2n+ 1, 0⟩. (2)

If y : o has measurement 0, then Iy →β y and the corresponding measurements are 1 > 0.

Example. Barendregt’s note [4] contains ten hand-evaluated examples. For the term E =
(λfo→o.λxo.fx)I, our Scheme computation in [10] gives 2, whereas the value printed in that
note is 1. By (2), [[I]]∅ = ⟨i, 0⟩ with i(n) = 2n+ 1. For the first contractum J = λxo.Ix, the
dot part is j(n) = addn+1(i(n)) = 3n + 2 and the star part is i(0) = 1. Applying the outer
abstraction of E to [[I]] adds [[I]]∗ + 1 = 1, giving

[[J ]]∅ = ⟨λn. 3n+ 2, 1⟩, [[E]]∅ = add1([[J ]]∅) = ⟨λn. 3n+ 3, 2⟩.

Thus the star components follow the full reduction sequence exactly in this case:

(λf.λx.fx)I →β λx.Ix →β I.
2 1 0

3 The Simple Haskell Program
The Haskell file is compact and uses only the standard base library. Running it with the example
outputs 2,3 the value calculated just above.

Types. An algebraic datatype is described by its constructors; a generalized algebraic datatype
(GADT) specifies more precise constructor result types [16]. The empty Haskell type Base
represents o, and the GADT Ty represents object types. The initial pragmas enable GADTs
and the empty marker type.

3runghc SN_Haskell.hs.

{-# LANGUAGE EmptyDataDecls #-}
{-# LANGUAGE GADTs #-}
{-# OPTIONS_GHC -Wall #-}

module Main where

import Numeric.Natural (Natural)

-- Object-language types, indexed by Haskell types.

data Base

data Ty a where
TBase :: Ty Base
TArr :: Ty a -> Ty b -> Ty (a -> b)

Measures. The datatype Meas mirrors the domains Dτ . Constructor N stores a base-type
natural number. Constructor F stores a Haskell function for the dot component and a natural
number for the star component. The definitions of star, dot, add, and canon are direct
translations of Section 2.

-- Type-indexed de Vrijer measurements.

data Meas a where
N :: Natural -> Meas Base
F :: (Meas a -> Meas b) -> Natural -> Meas (a -> b)

star :: Meas a -> Natural
star (N n) = n
star (F _ n) = n

dot :: Meas (a -> b) -> Meas a -> Meas b
dot (F f _) a = f a

add :: Natural -> Meas a -> Meas a
add k (N n) = N (n + k)
add k (F f n) = F (\a -> add k (f a)) (n + k)

canon :: Ty a -> Natural -> Meas a
canon TBase n = N n
canon (TArr _ b) n = F (\a -> canon b (n + star a)) n

GHC, the Glasgow Haskell Compiler, uses the indices of Ty (above) and Meas and the type
of app (below) to rule out ill-typed applications.

Terms. The line type Term a = Meas a is the key simplification. An object term is repre-
sented by its measurement directly. Variables are host-language variables, application is dot, and
abstraction is de Vrijer’s operation. The Haskell closure created by lam carries the environment
that our first-order implementation passes explicitly.

-- Higher-order/final representation of terms.
-- Object binders are Haskell binders, so Haskell closures carry the
-- environment that the first-order version passes explicitly.

type Term a = Meas a

var :: Meas a -> Term a
var = id

app :: Term (a -> b) -> Term a -> Term b
app = dot

lam :: Ty a -> (Term a -> Term b) -> Term (a -> b)
lam ty body =

F (\a -> add (star a + 1) (body a))
(star (body (canon ty 0)))

measure :: Term a -> Natural
measure = star

Example. The previous example is represented by ordinary Haskell application and binders:

-- Example: (\f^{o -> o}. \x^o. f x) (\x^o. x)

base :: Ty Base
base = TBase

baseToBase :: Ty (Base -> Base)
baseToBase = TArr base base

ex :: Term (Base -> Base)
ex = app (lam baseToBase (\f -> lam base (\x -> app f x)))

(lam base (\x -> x))

main :: IO ()
main = print (measure ex)

The two object-level binders printed as x in the mathematical notation are distinct Haskell
binders with separate lexical scopes. The value measure ex is 2. The checker in the archive
verifies the reported star values; when an example specifies a functional component, it compares
that function on the inputs 0..4.

The code also contains a first-order named version and a trace program.

4 Proof of Correctness
The proof follows the structure of the above construction. We first isolate the ordered, monotone
part of each measurement domain; then show that add, c, and term interpretation preserve it;
establish substitution; and finally lift root decrease through every reduction context.

Definition 1 (Orders and admissibility). For each type τ , define a non-strict quasi-order ≥τ , a
strict order >τ , and a predicate of admissibility on Dτ . At base type, the orders are the usual
orders on N, and every natural number is admissible. At function type, with both function
comparisons quantified over all admissible a ∈ Dσ,

⟨f,m⟩ ≥σ→τ ⟨g, n⟩ iff m ≥ n and f(a) ≥τ g(a),

⟨f,m⟩ >σ→τ ⟨g, n⟩ iff m > n and f(a) >τ g(a).

A function-type measure ⟨f, n⟩ ∈ Dσ→τ is admissible when f(a) is admissible for every admissible
a ∈ Dσ, and, for all admissible a, b ∈ Dσ, f is monotone for both comparisons: a ≥σ b ⇒
f(a) ≥τ f(b) and a >σ b ⇒ f(a) >τ f(b). A valuation is admissible when it maps each variable
to an admissible measure of the variable’s type. For valuations ρ and ρ′, write ρ ≥ ρ′ when
ρ(x) ≥type(x) ρ

′(x) for every variable x.

The two pointwise properties have distinct roles in contextual reduction. When the function
position decreases from ⟨f,m⟩ to ⟨g, n⟩, the definition of >σ→τ gives f(a) > g(a) at every
admissible argument a. When the argument position decreases from a to b, admissibility of the
unchanged function measurement gives f(a) > f(b). Admissibility therefore identifies the part
of the full function space that preserves reductions of its arguments.

An admissible erasing function. Let Ky = λxo.y, where y, z : o and ρ(y) = ρ(z) = 0.
Although the body ignores its argument, the abstraction clause gives [[Ky]]ρ = ⟨λn. n + 1, 0⟩,
whose dot part is strictly increasing. With I = λxo.x, the sequence Ky(Iz) →β Kyz →β y
has star values 2 > 1 > 0. Thus the charge for the argument records reductions that can take
place before that argument is discarded. If the abstraction clause added only 1, omitting the
argument’s star component, the first two measurements would both be 1 and strict decrease in
the argument position would fail.

A counterexample shows what would happen were every measurement declared admissible.
Let x : o → o and y : o, put ρ(y) = 0, and assign ρ(x) = ⟨h, 0⟩, where h(0) = 1 and
h(n + 1) = 0. For Q = (λzo.z)y, we have [[Q]]ρ = 1 > 0 = [[y]]ρ. Yet xQ →β xy while
[[xQ]]ρ = h(1) = 0 < 1 = h(0) = [[xy]]ρ. Thus h reverses the strict comparison and ⟨h, 0⟩ is not
admissible. Canonical measurements and denotations under admissible valuations satisfy the
conditions, as proved below.

The strict order is contained in the non-strict order, and a >τ b always implies a∗ > b∗.
It is therefore well founded on admissible measurements, since every strict descent lowers the
natural-number star component.

Next we relate our simple Haskell representation to de Vrijer’s equations. A Haskell valuation
η assigns a value of type Meas a to each source variable of the corresponding object type. As
for mathematical valuations, an unlisted variable x : σ receives its canonical zero measurement,
here canon σ̂ 0. The value represented by a term is defined recursively by

Hη(x) = var (η(x)),

Hη(MN) = app (Hη(M)) (Hη(N)),

Hη(λx
σ.M) = lam σ̂ (λp.Hη[x7→p](M)),

where σ̂ is the Haskell witness of the object type σ. Let ι : N → Natural be the canonical
representation of natural numbers in Haskell. Compare Haskell values with mathematical
measurements by the type-indexed relation

N z ≃o n ⇐⇒ z = ι n,

F f z ≃σ→τ ⟨g, n⟩ ⇐⇒ z = ι n and
(
p ≃σ a ⇒ f(p) ≃τ g(a)

)
for all p, a.

Write η ≃ ρ when corresponding entries of the two valuations are related in this sense.

Proposition 2 (Program implements the equations). For every well-typed term M : τ , every
typed mathematical valuation ρ, and every Haskell valuation η ≃ ρ, we have Hη(M) ≃τ [[M ]]ρ.
Consequently, for every closed term M ,

measure (H∅(M)) = ι
(
[[M ]]∗∅

)
. (3)

Proof. Induction on types shows that the two implementations of the auxiliary operations
respect ≃: if p ≃τ a, then add (ι k) p ≃τ addk(a); also canon τ̂ (ι n) ≃τ cτ (n). Now induct
structurally on M . The variable case is var = id, and the application case is app = dot. For
an abstraction, fix arbitrary related arguments p ≃σ a. The extended valuations η[x 7→ p] and
ρ[x 7→ a] correspond, so the induction hypothesis for the body, followed by compatibility of
add, gives the related dot components. Applying the same induction hypothesis at the related
canonical measurements gives the related star components. This proves the function-type clause
of ≃. Equation (3) follows from measure = star.

Some elementary facts are used repeatedly in the termination proof.

Lemma 3 (Auxiliary operations). Let τ be a type. For all admissible a, b ∈ Dτ and all k, ℓ ∈ N,

addk(a) is admissible, (4)
addk(a)

∗ = a∗ + k, (5)
a ≥τ b and k ≥ ℓ =⇒ addk(a) ≥τ addℓ(b), (6)

a ≥τ b, k ≥ ℓ, and (a >τ b or k > ℓ) =⇒ addk(a) >τ addℓ(b), (7)
k > 0 =⇒ addk(a) >τ a. (8)

Moreover, for all m,n ∈ N,

cτ (n) is admissible, (9)
cτ (n)

∗ = n, (10)
n ≥ m =⇒ cτ (n) ≥τ cτ (m), (11)
n > m =⇒ cτ (n) >τ cτ (m). (12)

Proof. Induct on τ . At base type, Equations (4)–(12) reduce to elementary arithmetic. Suppose
τ = σ → υ, and write a = ⟨f, r⟩ and b = ⟨g, s⟩. Equation (5) and the star part of Equations (6)
and (7) follow immediately from the definitions and the corresponding arithmetic comparisons
r + k ≥ s+ ℓ or r + k > s+ ℓ.

For Eq. (4), admissibility of a gives admissibility of f(c) for every admissible c; the induction
hypothesis for (4) then gives admissibility of addk(f(c)). For admissible c, d with c ≥ d,
admissibility of a gives f(c) ≥ f(d), and the induction hypothesis for (6), with the same k on
both sides, gives addk(f(c)) ≥ addk(f(d)). The strict-monotonicity clause follows in the same
way from the induction hypothesis for (7). Thus addk(a) is admissible.

For Eq. (6), the function-type hypothesis a ≥ b gives f(c) ≥ g(c) at every admissible c;
applying the induction hypothesis for (6) pointwise yields addk(f(c)) ≥ addℓ(g(c)). Equation (7)
is obtained pointwise from its induction hypothesis: strictness comes from f(c) > g(c) when
a > b, or from k > ℓ otherwise. Equation (8) follows from (7) with b = a and ℓ = 0, using the
reflexivity of ≥τ on admissible measurements and the identity add0(a) = a.

For canonical measurements, Eq. (10) is immediate. At an admissible argument c, the dot
component of cσ→υ(n) is cυ(n+ c∗), which is admissible by the induction hypothesis for (9). If
c ≥ d, then c∗ ≥ d∗, so the induction hypothesis for (11) gives non-strict monotonicity of the
dot component; if c > d, the corresponding star components are strictly ordered, and (12) gives
strict monotonicity. This proves (9). Finally, Equations (11) and (12) follow pointwise from
their induction hypotheses at υ, since addition of the common value c∗ preserves the comparison
between n and m.

Lemma 4 (Representation invariant). If M : τ and ρ is admissible, then [[M ]]ρ is an admissible
element of Dτ . In addition, [[M ]]ρ is monotone in ρ: if admissible valuations ρ ≥ ρ′ pointwise,
then [[M ]]ρ ≥τ [[M ]]ρ′ .

Proof. By induction on the typing derivation. For variables, lookup returns either the assigned
admissible value or a canonical one, which is admissible by (9). For an application MN ,
the induction hypothesis gives an admissible function measurement for M and an admissible
argument measurement for N ; admissibility of the function measurement gives admissibility of
[[MN ]]ρ. For monotonicity of an application, write [[M ]]ρ = ⟨f,m⟩, [[M ]]ρ′ = ⟨g, n⟩, a = [[N ]]ρ,
and b = [[N ]]ρ′ . The induction hypothesis gives ⟨f,m⟩ ≥ ⟨g, n⟩ and a ≥ b. Hence f(a) ≥ g(a) by
the function-type order and g(a) ≥ g(b) by admissibility of ⟨g, n⟩.

For an abstraction λxσ.M , the star component is a natural number by induction under
ρ[x 7→ cσ(0)]. For an admissible argument a, induction under ρ[x 7→ a] and Eq. (4) show
that adda∗+1([[M ]]ρ[x7→a]) is admissible. For admissible a ≥ b, we have ρ[x 7→ a] ≥ ρ[x 7→ b]
and a∗ + 1 ≥ b∗ + 1; Eq. (6) gives non-strict monotonicity of the dot part. If a > b, the
body comparison is non-strict and the numerical parameter is strict, so Eq. (7) gives strict
monotonicity. Monotonicity in ρ is obtained by applying the induction hypothesis to the body
under the corresponding extended valuations, both for the dot component and for the star
component.

Lemma 5 (Substitution). Let M and N be well typed, with type(N) = type(x), and suppose
substitution is capture-avoiding. Then, extensionally,

[[M [x 7→ N ]]]ρ = [[M ]]ρ[x7→[[N ]]ρ].

Proof. Induct on the structure of M , proving simultaneously the usual weakening property:
changing a valuation at a variable not free in a term does not change that term’s denotation. The
variable cases are the two lookup cases. The application case follows by applying the induction
hypotheses to both subterms. For an abstraction, α-convert first so that the binder y is different
from x and does not occur free in N . For the dot component, fix an arbitrary argument a,
extend both valuations with y 7→ a, and apply the induction hypothesis to the body. For the
star component, extend both valuations with y 7→ ctype(y)(0) and apply the induction hypothesis
again. Since y /∈ FV(N), weakening keeps [[N ]]ρ unchanged under the added y-binding, and the
two updates commute.

Lemma 6 (Reduction decreases the measure). If M →β N and both terms have type τ , then
every admissible valuation ρ satisfies [[M ]]ρ >τ [[N ]]ρ.

Proof. Induct on the reduction context. At the root, (λxσ.P )Q →β P [x 7→ Q]. Let a = [[Q]]ρ.
Equation (1) gives [[(λx.P )Q]]ρ = adda∗+1([[P ]]ρ[x7→a]). By Lemma 4, a is admissible, so ρ[x 7→ a]
is an admissible valuation; another use of Lemma 4 gives admissibility of [[P ]]ρ[x7→a]. By Lemma 5,
[[P ]]ρ[x7→a] = [[P [x 7→ Q]]]ρ. Since a∗ + 1 > 0, (8) gives the strict inequality for the contracted
redex.

For a reduction in the function position, PQ →β P ′Q with P →β P ′. The induction
hypothesis gives [[P ]]ρ > [[P ′]]ρ. Writing these function measurements as ⟨f,m⟩ and ⟨g, n⟩,
and setting a = [[Q]]ρ, Lemma 4 gives admissibility of a. The definition of >σ→τ then gives
f(a) > g(a), which is the required comparison of applications. For a reduction in the argument,
PQ →β PQ′, induction gives [[Q]]ρ > [[Q′]]ρ; Lemma 4 gives admissibility of [[P ]]ρ, so its dot part
is strictly monotone.

For a reduction below a lambda, λx.P →β λx.P ′, compare star components by applying
the induction hypothesis to the body under the admissible valuation ρ[x 7→ cσ(0)]. For each

admissible argument a, compare dot components by applying the induction hypothesis under
the admissible valuation ρ[x 7→ a], and then use (7) with equal numerical parameters to preserve
strictness under adda∗+1. Thus the abstraction measurements are strictly ordered.

Theorem 7 (Correctness of measure). For every closed, well-typed source term M , the value
measure (H∅(M)) is a natural number. If M →β N , then

measure (H∅(M)) > measure (H∅(N)). (13)

Therefore, the length of a β-reduction sequence starting from M is at most the initial value.

Proof. Equation (3) identifies the program’s output with the Natural representation of [[M ]]∗∅.
The empty valuation is admissible. If M →β N , subject reduction gives that N has the same
type as M , and N remains closed. Lemma 6 gives [[M ]]∅ > [[N ]]∅ and therefore [[M ]]∗∅ > [[N ]]∗∅.
With Eq. (3), this proves (13). The argument repeats at each subsequent step; every step lowers
the natural number by at least one, so the initial value bounds the length of the sequence.

Interpreting the output. For a closed, well-typed term M , ∥M∥ = 0 exactly when M
is β-normal. Under the canonical-zero valuation, a neutral normal form xN1 · · ·Nk : τ has
measurement cτ (0): each normal argument has star component 0. Simultaneous induction on
neutral and normal forms proves this claim and shows that an abstraction has star component 0
by applying the induction hypothesis to its body with the bound variable assigned its canonical
zero. Conversely, Lemma 6 gives a strictly positive star component to every term that admits a
reduction step. The positive output is a bound rather than necessarily the exact length. For
example, with I = λxo.x, the sequence λyo.I(Iy) →β λyo.Iy →β λyo.y has measures 3 > 1 > 0,
although every complete reduction sequence from the first term has length 2. De Vrijer also gives
an exact valuation whose star component equals the length of a longest β-reduction sequence [8,
Sections 2.3 and 4]. The present implementation uses his simpler loose valuation [8, Section 2.4].

The Rocq formalization. The file SN_Haskell_Rocq.v formalizes the measurement algebra,
typed syntax, substitution, contextual reduction, and the normalization argument above.4 Every
lemma and theorem in the development has a complete proof. Equality of function components
is handled extensionally using Rocq’s standard functional-extensionality principle.

The common measurement algebra is represented by Haskell’s Meas a and Rocq’s den t. The
operations star, dot, add, and canon have the same names in both programs; Rocq’s appD and
lamD implement Haskell’s app and lam. The type synonym Term a = Meas a makes the Haskell
representation final: constructing a term with the combinators computes its measurement. Rocq
instead folds explicit syntax into this algebra. Haskell’s Natural and Rocq’s nat correspond to
N through ι and ≃o.

Intrinsically typed de Bruijn syntax. The Rocq families var Gamma t and tm Gamma t
represent a de Bruijn variable and a term of type t in context Γ. The constructors TVar, TApp,
and TLam enforce typing in their indices. A valuation in Val Gamma assigns each typed index
a measurement of the corresponding type. The evaluator is the fold from this syntax into the
common measurement algebra:

evalρ(TVar i) = ρ(i),

evalρ(TAppM N) = appD
(
evalρ(M), evalρ(N)

)
,

evalρ(TLamM) = lamD
(
λa. evala::ρ(M)

)
.

4With Rocq 9.2.0 [23], run rocq compile SN_Haskell_Rocq.v.

Here a :: ρ is the Rocq operation val_ext, which assigns a to de Bruijn index 0 and shifts the
previous valuation to the remaining indices.

Fix an ordered context Γ = (x0 : τ0, . . . , xr−1 : τr−1) of distinct variables, with x0 represented
by de Bruijn index 0. After the usual α-renaming of binders, dbΓ maps xi to its typed index, maps
application componentwise, and maps λxσ.M to TLam(db(x:σ,Γ)(M)). For a named valuation ρ,
the indexed valuation ρ̄ maps the index of xi to ρ(xi); under a binder it is extended by val_ext.

For the running example, db∅(E) = (λo→o.λo.1 0)(λo.0). In the doubly nested body, indices
0 and 1 denote x and f , represented by TVar vz and TVar (vs vz). The resulting Rocq term
is ex_tm; ex_star proves that its star component is 2, agreeing with measure ex.

Proposition 8 (Haskell/de Bruijn correspondence). Let M : τ be well typed in Γ, let ρ be any
typed mathematical valuation, let η ≃ ρ be a corresponding Haskell valuation, and let ρ̄ be the
indexed Rocq valuation just defined. Then Hη(M) ≃τ [[M ]]ρ = evalρ̄

(
dbΓ(M)

)
, where the equality

uses the recursive identification of Dτ with the Rocq carrier den t. For a closed term, taking
star components gives measure (H∅(M)) = ι

(
star(eval∅(db∅(M)))

)
.

Proof. By Proposition 2, Hη(M) ≃τ [[M ]]ρ. It remains to prove [[M ]]ρ = evalρ̄(dbΓ(M)), which we
do by structural induction on M . A named variable and its de Bruijn index select corresponding
entries of ρ and ρ̄. Application uses the same dot operation on both sides. For an abstraction,
extend the two valuations by an arbitrary argument measurement and apply the induction
hypothesis to the body; apply it again at cσ(0) for the star component. The clauses for lamD and
the mathematical abstraction operation then agree, extensionally at arrow type. The closed-term
equation follows by taking star components and using Eq. (3).

Proposition 8 is the structural bridge. Starting from de Bruijn syntax, Rocq proves the
representation invariant in eval_adm_ge and substitution in eval_subst0, whose equation is
evalρ(subst0 N M) = evalevalρ(N)::ρ(M). The four constructors of step cover root contraction,
either application position, and reduction under a lambda. The theorem step_decreases
proves Lemma 6; star_decreases projects it to natural numbers. Well-founded induction on
this component gives accessibility in strong_normalization; strong_normalization_all uses
canonical valuations and strong_normalization_closed specializes to closed terms. The length
bound follows by iterating one-step descent. A declaration-by-declaration table accompanies the
sources as CORRESPONDENCE.md.

5 Conclusion
Our Haskell implementation follows de Vrijer’s construction, using host-language functions for
dot components and binders for the object-language environment. This eliminates the explicit
environment parameter and the quoting and unquoting of our earlier Scheme implementation [10].
The proof establishes that the computed star component is in fact a decreasing natural-number
bound for full simply typed β-reduction. The complementary Rocq evaluator folds intrinsically
typed de Bruijn syntax into the same algebra and formalizes substitution, contextual decrease,
and strong normalization.

The archive for this work includes the simple higher-order Haskell program, named and
tracing versions, an example checker, and the Rocq development.

